clear all
set more off
set mem 10000000
set matsize 10000
version 15

***************************************************************************************
*** RDROBUST first stage, nighttime brightness: splits on RGGVY treatment intensity ***
***************************************************************************************

** Set file paths
do "$path_code/paths.do"

** Set graph scheme
cd "$path/code/analyze"
set scheme fb, perm

****************************************************************** 
****************************************************************** 

{
use "$panel/panel_dataset_full.dta", clear

	// Bring in RGGVY district-level administrative data (10th Plan only)
preserve
use "$rggvy/rggvy_district_progress_X_XI_processed.dta", clear
egen dt_group = group(st_code dt_code)
egen temp_group = group(plan implement_type)
unique st_code, by(temp_group) gen(uniq_st)
unique dt_group, by(temp_group) gen(uniq_dt)
unique dpr_code, by(temp_group) gen(uniq_dpr)
unique st_code if plan==10
unique st_code if plan==11
unique dt_group if plan==10
unique dt_group if plan==11
count if award_date<sanction_date
replace award_date = sanction_date if award_date<sanction_date
replace award_date = max(award_date,17553) if plan==11
keep if plan==10
collapse (min) min_award_date=award_date (max) max_award_date=award_date ///
	(sum) award_cost total_released achiev_UDE achiev_ELEC achiev_BPL ///
	, by(st_code dt_code)	
gen med_award_date = round((min_award_date+max_award_date)/2,1)
format %td med_award_date	
tempfile admin
save `admin'
restore	
merge m:1 st_code dt_code using `admin' 

	// RGGVY admin splitter 1: (# villages treated) / (# villages in district)
egen temp1 = count(vi_code) if tot_p>=300, by(st_code dt_code)
egen temp2 = mode(temp1), by(st_code dt_code)
gen RGGVY_share_villages_300 = (achiev_UDE + achiev_ELEC) / temp2
egen temp3 = count(vi_code), by(st_code dt_code)
gen RGGVY_share_villages_all = (achiev_UDE + achiev_ELEC) / temp3
egen temp_tag = tag(st_code dt_code) 
twoway scatter RGGVY_share_villages_300 RGGVY_share_villages_all if temp_tag
tab state if temp_tag & RGGVY_share_villages_300>1.4 & RGGVY_share_villages_all!=. & corr_state==1
gen RGGVY_rule_breaker = RGGVY_share_villages_300>1.4
twoway scatter RGGVY_share_villages_300 RGGVY_share_villages_all if temp_tag & RGGVY_rule_breaker==0
sum RGGVY_share_villages_all if temp_tag & RGGVY_rule_breaker==0, detail
sum RGGVY_share_villages_300 if temp_tag & RGGVY_rule_breaker==0, detail
// split on 60% of villages in district
drop temp*

	// RGGVY admin splitter 2: (# BPL HHs treated) / (# village HHs in district)
egen temp1 = sum(no_hh11) if tot_p>=300, by(st_code dt_code)
egen temp2 = mode(temp1), by(st_code dt_code)
gen RGGVY_share_hh_300 = (achiev_BPL) / temp2
egen temp3 = sum(no_hh11), by(st_code dt_code)
gen RGGVY_share_hh_all = (achiev_BPL) / temp3
egen temp_tag = tag(st_code dt_code) 
twoway scatter RGGVY_share_hh_300 RGGVY_share_hh_all if temp_tag 
twoway scatter RGGVY_share_hh_300 RGGVY_share_hh_all if temp_tag & RGGVY_rule_breaker==0
sum RGGVY_share_hh_all if temp_tag & RGGVY_rule_breaker==0, detail
sum RGGVY_share_hh_300 if temp_tag & RGGVY_rule_breaker==0, detail
// split on 10% of HHs in district
drop temp*

	// RGGVY admin splitter 3: (Rs allocated) / (# villages treated)
egen temp_tag = tag(st_code dt_code) 
gen RGGVY_lakh_per_v = (total_released) /  (achiev_UDE + achiev_ELEC)
sum RGGVY_lakh_per_v if temp_tag & RGGVY_rule_breaker==0, detail
// split on 10 lakh per village
drop temp*
	
	// Keep villages in RD sample
gen in_fs_sample = vplan4<11 & corr_state==1 & sample==1 & sample_h==1 & sing_h==1 & pop_non_zero==1 
keep if in_fs_sample==1
	
	// Create state FEs (since RD robust doesn't let you pass them through)
drop if st_code==32 // Kerala, only 3 villages
tab st_code, gen(STFE)	
drop STFE1 // to avoid collinearity

	// Create district FEs (since RD robust doesn't let you pass them through)
gen stdtFE = stdt
tab stdtFE state     if inlist(stdtFE,62,63,64,70,91,180,181,202,308,309,320,490,493,494,495,504,505,508)	
replace stdtFE = 999 if inlist(stdtFE,62,63,64,70,91,180,181,202,308,309,320,490,493,494,495,504,505,508)	
	// one catch-all district FE for districts with so few in-sample villages that they break rdrobust
tab stdtFE, gen(DTFE)	
drop DTFE1 // to avoid collinearity

	// Create block groups (for clustering)
egen stdtbk = group(stdt bk_code)

	// Create lights-difference variable, to identify crazy outliers
gen lights_diff = abs(lights_max2011_hat - lights_max2001_hat)


  // Create variables to store regresson results
gen fs_step = .
gen dpr_step = .
gen yvar = ""
gen ifs = ""
gen dpr_if = ""
gen control = ""
gen fe = ""
gen kernel = ""
gen bwmethod = ""
gen vce = ""
gen polynomial_order = .
gen beta_conv = .
gen beta_robust = .
gen se_conv = .
gen se_robust = .
gen pval_conv = .
gen pval_robust = .
gen lci_conv = .
gen uci_conv = .
gen lci_robust = .
gen uci_robust = .
gen bw_lo = .
gen bw_hi = .
gen nobs_orig = .
gen nobs_left = .
gen nobs_right = .
gen ndist = . 
gen ymean = .
gen ftag = ""
gen dprtag = ""
local folder = "RDROBUST plots fs lights DPR"
local h_wide = 200


	// 1.1 Lights brightness regressions, preferred
local fs_step = 1
local ifs = "pop_mismatch20==0 & lights_diff<20"
local CONTROLS = "lights_max1998 lights_max1999 lights_max2000 lights_max2001 lights_max2002 lights_max2003 lights_max2004 lights_max2005" 
local fe = "STFE*"
local kernel = "tri"
local bwmethod = "mserd"
local vce = ""
local poly = 1
local row = 0

foreach dpr in 1 2 3 4 5 6 7 {
	
	if `dpr'==1 {
		local dpr_if = " & RGGVY_rule_breaker==0"
		local dprtag = "rule"
	}
	else if `dpr'==2 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_share_villages_all>=0.6"
		local dprtag = "hi_vill"
    }
	else if `dpr'==3 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_share_villages_all<0.6"
		local dprtag = "lo_vill"
    }
	else if `dpr'==4 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_share_hh_all>=0.1"
		local dprtag = "hi_hh"
    }
	else if `dpr'==5 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_share_hh_all<0.1"
		local dprtag = "lo_hh"
    }
	else if `dpr'==6 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_lakh_per_v>=10"
		local dprtag = "hi_funds"
    }
	else if `dpr'==7 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_lakh_per_v<10"
		local dprtag = "lo_funds"
    }
   
	foreach y in 2002 2003 2004 2005 2006 2007 2008 2009 2010 2011 2012 {

		// Define outcome variable
		local yvar = "lights_max`y'_hat"
		local title = "Preferred specification: `y' Brightness"
		local ftag = "preferred"

		// Remove controls used to project outcome variable
		local controls = "`CONTROLS'"
		if regexm("`yvar'","2007")==1 {
			local controls = subinstr("`controls'"," lights_max2005","",1)
		}
		else if regexm("`yvar'","2006")==1 {
			local controls = subinstr("`controls'"," lights_max2004 lights_max2005","",1)
		}
		else if regexm("`yvar'","2005")==1 {
			local controls = subinstr("`controls'"," lights_max2003 lights_max2004 lights_max2005","",1)
		}
		else if regexm("`yvar'","2004")==1 {
			local controls = subinstr("`controls'"," lights_max2002 lights_max2003 lights_max2004 lights_max2005","",1)
		}
		else if regexm("`yvar'","2003")==1 {
			local controls = subinstr("`controls'"," lights_max2001 lights_max2002 lights_max2003 lights_max2004 lights_max2005","",1)
		}
		else if regexm("`yvar'","2002")==1 {
			local controls = subinstr("`controls'"," lights_max2000 lights_max2001 lights_max2002 lights_max2003 lights_max2004 lights_max2005","",1)
		}
		
		// Run first-stage regresssion
		rdrobust `yvar' tot_p if `ifs' `dpr_if', c(299.5) covs(`controls' `fe') kernel(`kernel') bwselect(`bwmethod') p(`poly') all `vce'

		// Generate in-sample indicator and store bandwidths
		cap drop temp_in_reg
		qui gen temp_in_reg = `ifs' `dpr_if' & inrange(tot_p,299.5-e(h_l),299.5+e(h_r))
		qui count if temp_in_reg
		assert r(N)==(e(N_h_l)+e(N_h_r))
		local h_l = e(h_l)
		local h_r = e(h_r)

		// Store results
		local row = `row' + 1
		qui replace fs_step = `fs_step' in `row'
		qui replace dpr_step = `dpr' in `row'
		qui replace yvar = "`yvar'" in `row'
		qui replace ifs = "`ifs'" in `row'
		qui replace dpr_if = "`dpr_if'" in `row'
		qui replace control = "`controls'" in `row'
		qui replace fe = "`fe'" in `row'
		qui replace kernel = e(kernel) in `row'
		qui replace bwmethod = e(bwselect) in `row'
		qui replace vce = "`vce'" in `row'
		qui replace polynomial_order = e(p) in `row'
		qui replace beta_conv = e(tau_cl) in `row'
		qui replace beta_robust = e(tau_bc) in `row'
		qui replace se_conv = e(se_tau_cl) in `row'
		qui replace se_robust = e(se_tau_rb) in `row'
		qui replace pval_conv = e(pv_cl) in `row'
		qui replace pval_robust = e(pv_rb) in `row'
		qui replace lci_conv = e(ci_l_cl) in `row'
		qui replace uci_conv = e(ci_r_cl) in `row'
		qui replace lci_robust = e(ci_l_rb) in `row'
		qui replace uci_robust = e(ci_r_rb) in `row'
		qui replace bw_lo = e(h_l) in `row'
		qui replace bw_hi = e(h_r) in `row'
		qui replace nobs_orig = e(N) in `row'
		qui replace nobs_left = e(N_h_l) in `row'
		qui replace nobs_right = e(N_h_r) in `row'
		qui unique stdt if temp_in_reg==1
		qui replace ndist = r(unique) in `row'
		qui sum `yvar' if temp_in_reg==1 & tot_p<299.5
		qui replace ymean = r(mean) in `row'
		qui replace ftag = "`ftag'" in `row'
		qui replace dprtag = "`dprtag'" in `row'

		// Residualize `yvar' for regression sample
		qui reg `yvar' `controls' `fe' if temp_in_reg==1
		cap drop y_resid_sample
		predict y_resid_sample, residuals
		
		// Residualize `yvar' for [100,500] bandwidth
		qui reg `yvar' `controls' `fe' if `ifs' `dpr_if' & inrange(tot_p,299.5-`h_wide',299.5+`h_wide')
		cap drop y_resid_full
		predict y_resid_full, residuals
		
		// RD plots for regression sample (variable bandwidth) 
		foreach bins in 10 20 40 {
			if `h_l'<=75 {
				local xlab = "250 275 300 325 350"
			}
			else if inrange(`h_l',76,125) {
				local xlab = "200 250 300 350 400"
			}
			else if inrange(`h_l',126,175) {
				local xlab = "150 200 250 300 350 400 450"
			}
			else {
				local xlab = ""
			}
			rdplot y_resid_sample tot_p if temp_in_reg==1, ///
				c(299.5) p(`poly') nbins(`bins' `bins') kernel(`kernel') h(`h_l' `h_r') ///
				graph_options( ///
				title("`title'", color(black) size(large)) ///
				ytitle("`y' brightness residuals", size(medlarge)) ///
				xtitle("2001 village population", size(medlarge)) ///
				ylabel(,nogrid angle(0) labsize(medlarge)) ///
				xlabel(`xlab', labsize(medlarge)) ///
				graphregion(color(white)) plotregion(fcolor(white)) graphregion(lcolor(white)) ///
				legend(off))
			graph export "$results/`folder'/lights_fs_`ftag'_`dprtag'_`y'_inreg_`bins'.pdf", replace
		}

		// RD plots for regression sample (constant bandwidth) 
		foreach bins in 10 20 40 {
			if `bins'==40 {
				local ymarks = "-.4 -.2 0 .2 .4"
			}
			else {
				local ymarks = "-.3 -.2 -.1 0 .1 .2 .3"
			}
			rdplot y_resid_full tot_p if `ifs' `dpr_if' & inrange(tot_p,299.5-`h_wide',299.5+`h_wide'), ///
				c(299.5) p(`poly') nbins(`bins' `bins') kernel(`kernel') h(`h_wide' `h_wide') ///
				graph_options( ///
				title("`title'", color(black) size(large)) ///
				ytitle("`y' brightness residuals", size(medlarge)) ///
				xtitle("2001 village population", size(medlarge)) ///
				ylabel(`ymarks',nogrid angle(0) labsize(medlarge)) ///
				xlabel(, labsize(medlarge)) ///
				graphregion(color(white)) plotregion(fcolor(white)) graphregion(lcolor(white)) ///
				legend(off))
			graph export "$results/`folder'/lights_fs_`ftag'_`dprtag'_`y'_wide_`bins'.pdf", replace
			local ymarks = ""
		}			
	}
}

	// 1.7 Sensitivity to different bandwidth calculations
local fs_step = 7
local ifs = "pop_mismatch20==0 & lights_diff<20"
local CONTROLS = "lights_max1998 lights_max1999 lights_max2000 lights_max2001 lights_max2002 lights_max2003 lights_max2004 lights_max2005" 
local fe = "STFE*"
local kernel = "tri"
local vce = ""
local poly = 1

foreach dpr in 1 2 3 4 5 6 7 {
	
	if `dpr'==1 {
		local dpr_if = " & RGGVY_rule_breaker==0"
		local dprtag = "rule"
	}
	else if `dpr'==2 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_share_villages_all>=0.6"
		local dprtag = "hi_vill"
    }
	else if `dpr'==3 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_share_villages_all<0.6"
		local dprtag = "lo_vill"
    }
	else if `dpr'==4 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_share_hh_all>=0.1"
		local dprtag = "hi_hh"
    }
	else if `dpr'==5 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_share_hh_all<0.1"
		local dprtag = "lo_hh"
    }
	else if `dpr'==6 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_lakh_per_v>=10"
		local dprtag = "hi_funds"
    }
	else if `dpr'==7 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_lakh_per_v<10"
		local dprtag = "lo_funds"
    }
   
	foreach y in 2002 2003 2004 2005 2006 2007 2008 2009 2010 2011 2012 {

		foreach reg in 1 2 3 4 5 {

			// Define outcome variable
			local yvar = "lights_max`y'_hat"

			// Define bandwidth method
			if `reg'==1 {
				local bwmethod = "msetwo"
				local title = "`y' Brightness, Bandwidth Method `bwmethod'"
				local ftag = "BW`bwmethod'"
			}
			if `reg'==2 {
				local bwmethod = "msesum"
				local title = "`y' Brightness, Bandwidth Method `bwmethod'"
				local ftag = "BW`bwmethod'"
			}
			if `reg'==3 {
				local bwmethod = "cerrd"
				local title = "`y' Brightness, Bandwidth Method `bwmethod'"
				local ftag = "BW`bwmethod'"
			}
			if `reg'==4 {
				local bwmethod = "certwo"
				local title = "`y' Brightness, Bandwidth Method `bwmethod'"
				local ftag = "BW`bwmethod'"
			}
			if `reg'==5 {
				local bwmethod = "cersum"
				local title = "`y' Brightness, Bandwidth Method `bwmethod'"
				local ftag = "BW`bwmethod'"
			}

			
			// Remove controls used to project outcome variable
			local controls = "`CONTROLS'"
			if regexm("`yvar'","2007")==1 {
				local controls = subinstr("`controls'"," lights_max2005","",1)
			}
			else if regexm("`yvar'","2006")==1 {
				local controls = subinstr("`controls'"," lights_max2004 lights_max2005","",1)
			}
			else if regexm("`yvar'","2005")==1 {
				local controls = subinstr("`controls'"," lights_max2003 lights_max2004 lights_max2005","",1)
			}
			else if regexm("`yvar'","2004")==1 {
				local controls = subinstr("`controls'"," lights_max2002 lights_max2003 lights_max2004 lights_max2005","",1)
			}
			else if regexm("`yvar'","2003")==1 {
				local controls = subinstr("`controls'"," lights_max2001 lights_max2002 lights_max2003 lights_max2004 lights_max2005","",1)
			}
			else if regexm("`yvar'","2002")==1 {
				local controls = subinstr("`controls'"," lights_max2000 lights_max2001 lights_max2002 lights_max2003 lights_max2004 lights_max2005","",1)
			}
			
			// Run first-stage regresssion
			rdrobust `yvar' tot_p if `ifs' `dpr_if', c(299.5) covs(`controls' `fe') kernel(`kernel') bwselect(`bwmethod') p(`poly') all `vce'

			// Generate in-sample indicator and store bandwidths
			cap drop temp_in_reg
			qui gen temp_in_reg = `ifs' `dpr_if' & inrange(tot_p,299.5-e(h_l),299.5+e(h_r))
			qui count if temp_in_reg
			assert r(N)==(e(N_h_l)+e(N_h_r))
			local h_l = e(h_l)
			local h_r = e(h_r)
			
			// Store results
			local row = `row' + 1
			qui replace fs_step = `fs_step' in `row'
			qui replace dpr_step = `dpr' in `row'
			qui replace yvar = "`yvar'" in `row'
			qui replace ifs = "`ifs'" in `row'
			qui replace dpr_if = "`dpr_if'" in `row'
			qui replace control = "`controls'" in `row'
			qui replace fe = "`fe'" in `row'
			qui replace kernel = e(kernel) in `row'
			qui replace bwmethod = e(bwselect) in `row'
			qui replace vce = "`vce'" in `row'
			qui replace polynomial_order = e(p) in `row'
			qui replace beta_conv = e(tau_cl) in `row'
			qui replace beta_robust = e(tau_bc) in `row'
			qui replace se_conv = e(se_tau_cl) in `row'
			qui replace se_robust = e(se_tau_rb) in `row'
			qui replace pval_conv = e(pv_cl) in `row'
			qui replace pval_robust = e(pv_rb) in `row'
			qui replace lci_conv = e(ci_l_cl) in `row'
			qui replace uci_conv = e(ci_r_cl) in `row'
			qui replace lci_robust = e(ci_l_rb) in `row'
			qui replace uci_robust = e(ci_r_rb) in `row'
			qui replace bw_lo = e(h_l) in `row'
			qui replace bw_hi = e(h_r) in `row'
			qui replace nobs_orig = e(N) in `row'
			qui replace nobs_left = e(N_h_l) in `row'
			qui replace nobs_right = e(N_h_r) in `row'
			qui unique stdt if temp_in_reg==1
			qui replace ndist = r(unique) in `row'
			qui sum `yvar' if temp_in_reg==1 & tot_p<299.5
			qui replace ymean = r(mean) in `row'
			qui replace ftag = "`ftag'" in `row'
			qui replace dprtag = "`dprtag'" in `row'

			// Residualize `yvar' for regression sample
			qui reg `yvar' `controls' `fe' if temp_in_reg==1
			cap drop y_resid_sample
			predict y_resid_sample, residuals
			
			// RD plots for regression sample (variable bandwidth) 
			foreach bins in 10 20 40 {
				if `h_l'<=75 {
					local xlab = "250 275 300 325 350"
				}
				else if inrange(`h_l',76,125) {
					local xlab = "200 250 300 350 400"
				}
				else if inrange(`h_l',126,175) {
					local xlab = "150 200 250 300 350 400 450"
				}
				else {
					local xlab = ""
				}
				rdplot y_resid_sample tot_p if temp_in_reg==1, ///
					c(299.5) p(`poly') nbins(`bins' `bins') kernel(`kernel') h(`h_l' `h_r') ///
					graph_options( ///
					title("`title'", color(black) size(large)) ///
					ytitle("`y' brightness residuals", size(medlarge)) ///
					xtitle("2001 village population", size(medlarge)) ///
					ylabel(,nogrid angle(0) labsize(medlarge)) ///
					xlabel(`xlab', labsize(medlarge)) ///
					graphregion(color(white)) plotregion(fcolor(white)) graphregion(lcolor(white)) ///
					legend(off))
				graph export "$results/`folder'/lights_fs_`ftag'_`dprtag'_`y'_inreg_`bins'.pdf", replace
			}
		}		
	}
}


	// 1.8 Sensitivity to 2nd-order polynomial and differet kernels
local fs_step = 8
local ifs = "pop_mismatch20==0 & lights_diff<20"
local CONTROLS = "lights_max1998 lights_max1999 lights_max2000 lights_max2001 lights_max2002 lights_max2003 lights_max2004 lights_max2005" 
local fe = "STFE*"
local bwmethod = "mserd"
local vce = ""

foreach dpr in 1 2 3 4 5 6 7 {
	
	if `dpr'==1 {
		local dpr_if = " & RGGVY_rule_breaker==0"
		local dprtag = "rule"
	}
	else if `dpr'==2 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_share_villages_all>=0.6"
		local dprtag = "hi_vill"
    }
	else if `dpr'==3 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_share_villages_all<0.6"
		local dprtag = "lo_vill"
    }
	else if `dpr'==4 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_share_hh_all>=0.1"
		local dprtag = "hi_hh"
    }
	else if `dpr'==5 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_share_hh_all<0.1"
		local dprtag = "lo_hh"
    }
	else if `dpr'==6 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_lakh_per_v>=10"
		local dprtag = "hi_funds"
    }
	else if `dpr'==7 {
		local dpr_if = " & RGGVY_rule_breaker==0 & RGGVY_lakh_per_v<10"
		local dprtag = "lo_funds"
    }
   
	foreach y in 2002 2003 2004 2005 2006 2007 2008 2009 2010 2011 2012 {

		foreach reg in 2 4 {

			// Define outcome variable
			local yvar = "lights_max`y'_hat"

			// Define kernel and polynomial combo
			if `reg'==1 {
				local kernel = "tri"
				local poly = 2
				local title = "`y' Brightness, Triangular Kernel, Quadratic"
				local ftag = "K`kernel'P`poly'"
			}
			if `reg'==2 {
				local kernel = "epa"
				local poly = 1
				local title = "`y' Brightness, Epanechnikov Kernel"
				local ftag = "K`kernel'P`poly'"
			}
			if `reg'==3 {
				local kernel = "epa"
				local poly = 2
				local title = "`y' Brightness, Epanechnikov Kernel, Quadratic"
				local ftag = "K`kernel'P`poly'"
			}
			if `reg'==4 {
				local kernel = "uni"
				local poly = 1
				local title = "`y' Brightness, Uniform Kernel"
				local ftag = "K`kernel'P`poly'"
			}
			if `reg'==5 {
				local kernel = "uni"
				local poly = 2
				local title = "`y' Brightness, Uniform Kernel, Quadratic"
				local ftag = "K`kernel'P`poly'"
			}

			
			// Remove controls used to project outcome variable
			local controls = "`CONTROLS'"
			if regexm("`yvar'","2007")==1 {
				local controls = subinstr("`controls'"," lights_max2005","",1)
			}
			else if regexm("`yvar'","2006")==1 {
				local controls = subinstr("`controls'"," lights_max2004 lights_max2005","",1)
			}
			else if regexm("`yvar'","2005")==1 {
				local controls = subinstr("`controls'"," lights_max2003 lights_max2004 lights_max2005","",1)
			}
			else if regexm("`yvar'","2004")==1 {
				local controls = subinstr("`controls'"," lights_max2002 lights_max2003 lights_max2004 lights_max2005","",1)
			}
			else if regexm("`yvar'","2003")==1 {
				local controls = subinstr("`controls'"," lights_max2001 lights_max2002 lights_max2003 lights_max2004 lights_max2005","",1)
			}
			else if regexm("`yvar'","2002")==1 {
				local controls = subinstr("`controls'"," lights_max2000 lights_max2001 lights_max2002 lights_max2003 lights_max2004 lights_max2005","",1)
			}
			
			// Run first-stage regresssion
			rdrobust `yvar' tot_p if `ifs' `dpr_if', c(299.5) covs(`controls' `fe') kernel(`kernel') bwselect(`bwmethod') p(`poly') all `vce'

			// Generate in-sample indicator and store bandwidths
			cap drop temp_in_reg
			qui gen temp_in_reg = `ifs' `dpr_if' & inrange(tot_p,299.5-e(h_l),299.5+e(h_r))
			qui count if temp_in_reg
			assert r(N)==(e(N_h_l)+e(N_h_r))
			local h_l = e(h_l)
			local h_r = e(h_r)
			
			// Store results
			local row = `row' + 1
			qui replace fs_step = `fs_step' in `row'
			qui replace dpr_step = `dpr' in `row'
			qui replace yvar = "`yvar'" in `row'
			qui replace ifs = "`ifs'" in `row'
			qui replace dpr_if = "`dpr_if'" in `row'
			qui replace control = "`controls'" in `row'
			qui replace fe = "`fe'" in `row'
			qui replace kernel = e(kernel) in `row'
			qui replace bwmethod = e(bwselect) in `row'
			qui replace vce = "`vce'" in `row'
			qui replace polynomial_order = e(p) in `row'
			qui replace beta_conv = e(tau_cl) in `row'
			qui replace beta_robust = e(tau_bc) in `row'
			qui replace se_conv = e(se_tau_cl) in `row'
			qui replace se_robust = e(se_tau_rb) in `row'
			qui replace pval_conv = e(pv_cl) in `row'
			qui replace pval_robust = e(pv_rb) in `row'
			qui replace lci_conv = e(ci_l_cl) in `row'
			qui replace uci_conv = e(ci_r_cl) in `row'
			qui replace lci_robust = e(ci_l_rb) in `row'
			qui replace uci_robust = e(ci_r_rb) in `row'
			qui replace bw_lo = e(h_l) in `row'
			qui replace bw_hi = e(h_r) in `row'
			qui replace nobs_orig = e(N) in `row'
			qui replace nobs_left = e(N_h_l) in `row'
			qui replace nobs_right = e(N_h_r) in `row'
			qui unique stdt if temp_in_reg==1
			qui replace ndist = r(unique) in `row'
			qui sum `yvar' if temp_in_reg==1 & tot_p<299.5
			qui replace ymean = r(mean) in `row'
			qui replace ftag = "`ftag'" in `row'
			qui replace dprtag = "`dprtag'" in `row'

			// Residualize `yvar' for regression sample
			qui reg `yvar' `controls' `fe' if temp_in_reg==1
			cap drop y_resid_sample
			predict y_resid_sample, residuals
			
			// Residualize `yvar' for [100,500] bandwidth
			qui reg `yvar' `controls' `fe' if `ifs' `dpr_if' & inrange(tot_p,299.5-`h_wide',299.5+`h_wide')
			cap drop y_resid_full
			predict y_resid_full, residuals
			
			// RD plots for regression sample (variable bandwidth) 
			foreach bins in 10 20 40 {
				if `h_l'<=75 {
					local xlab = "250 275 300 325 350"
				}
				else if inrange(`h_l',76,125) {
					local xlab = "200 250 300 350 400"
				}
				else if inrange(`h_l',126,175) {
					local xlab = "150 200 250 300 350 400 450"
				}
				else {
					local xlab = ""
				}
				rdplot y_resid_sample tot_p if temp_in_reg==1, ///
					c(299.5) p(`poly') nbins(`bins' `bins') kernel(`kernel') h(`h_l' `h_r') ///
					graph_options( ///
					title("`title'", color(black) size(large)) ///
					ytitle("`y' brightness residuals", size(medlarge)) ///
					xtitle("2001 village population", size(medlarge)) ///
					ylabel(,nogrid angle(0) labsize(medlarge)) ///
					xlabel(`xlab', labsize(medlarge)) ///
					graphregion(color(white)) plotregion(fcolor(white)) graphregion(lcolor(white)) ///
					legend(off))
				graph export "$results/`folder'/lights_fs_`ftag'_`dprtag'_`y'_inreg_`bins'.pdf", replace
			}

			// RD plots for regression sample (constant bandwidth) 
			foreach bins in 10 20 40 {
				rdplot y_resid_full tot_p if `ifs' `dpr_if' & inrange(tot_p,299.5-`h_wide',299.5+`h_wide'), ///
					c(299.5) p(`poly') nbins(`bins' `bins') kernel(`kernel') h(`h_wide' `h_wide') ///
					graph_options( ///
					title("`title'", color(black) size(large)) ///
					ytitle("`y' brightness residuals", size(medlarge)) ///
					xtitle("2001 village population", size(medlarge)) ///
					ylabel(,nogrid angle(0) labsize(medlarge)) ///
					xlabel(, labsize(medlarge)) ///
					graphregion(color(white)) plotregion(fcolor(white)) graphregion(lcolor(white)) ///
					legend(off))
				graph export "$results/`folder'/lights_fs_`ftag'_`dprtag'_`y'_wide_`bins'.pdf", replace
			}
		}		
	}
}


	// Save results
keep fs_step-dprtag
dropmiss, obs force
compress
save "$results/RDROBUST_fs_lights_dpr.dta", replace
}

****************************************************************** 
****************************************************************** 








